
Problem description: 
This example represents the bending of a thick beam in 2D plane strain conditions. 
The beam is clamped at the left end and loaded by concentrated forces at the right 
end. The material is linear elastic. 
Two solutions are obtained. The first one without domain decomposition is used as a 
reference, while the second uses various alternative domain decompositions, one of 
which is modal. 
Numerical Solutions 
DOMA00 
This calculation uses no domain decomposition and is used to obtain a reference 
solution. It uses a regular mesh of 24 elements of type CAR4. 
The mesh generation file is: 
 
*%siz 50 
* 
OPTI ECHO 1 DIME 2 ELEM QUA4; 
opti titr 'DOMA - 00'; 
opti trac psc ftra 'doma00_mesh.ps'; 
* 
P11=0. 0.; 
P12=0. 1.; 
P13=2. 0.; 
P14=2. 1.; 
* 
VEC1=2. 0.; 
* 
P1=P11; 
R1=P13; 
S1=P13 PLUS VEC1; 
* 
N2=3; 
N3=4; 

* 
P11_P12=P11 D N2 P12; 
P11_P12P=P11_P12 PLUS VEC1; 
P13_P14=P13 D N2 P14; 
P13_P14P=P13_P14 PLUS VEC1; 
* 
DOM11=P11_P12 REGL N3 P11_P12P; 
DOM12=P13_P14 REGL N3 P13_P14P; 
* 
TRAC (DOM11 ET DOM12); 
* 
ELIM 0.001 (DOM11 ET DOM12 ET P1 ET R1 ET S1); 
* 
TOUT=DOM11 ET DOM12; 
* 
OPTI SAUV FORMAT 'doma00.msh'; 
SAUV FORMAT TOUT; 
* 
FIN;

 
The input file is: 
 
DOMA - 00 
*----------------------------------------------------------------------- 
ECHO 
!conv win 
CAST tout 
*-----------------------------------------------------------Problem type 
DPLA NONL 
*-----------------------------------------------------------Dimensioning 
DIME 
  PT2L 36 CAR4 24 
  TABL 1 4 ECRO 522 
  FORC 10  MTPO 9 
TERM 
*---------------------------------------------------------------Geometry 
GEOM CAR4 tout  
     TERM 
*--------------------------------------------------Geometric Complements 

COMP GROU 1 
     'prob1'  LECT dom11 dom12 TERM 
     NGRO 3 
     'tpln'   LECT p1 r1 s1 TERM 
     'char2'  LECT p13_p14p TERM 
     'bloqa'  LECT p11_p12  TERM  
     COUL turq LECT prob1 TERM 
*----------------------------------------------------------Material data 
MATE 
   LINE RO 7800. YOUN 210.E9 NU 0.3 
        LECT tout TERM 
*--------------------------------------------------------Applied "loads" 
CHAR 1 FACT 2  
   FORC 2 0.75E7 LECT char2 TERM 
   TABL 2 0.0 1.0 1.0 1.0 
*----------------------------------------------------Boundary conditions 
LINK COUP BLOQ 12 LECT bloqa TERM 

*----------------------------------------------------------------Outputs 
ECRI DEPL TFRE 4D-3 NOEL  
          POIN LECT tpln TERM 
     FICH ALIC TEMP TFREQ 0.2D-3 
          POIN LECT tpln TERM 
          ELEM LECT 1 TERM 
     FICH ALIC TFRE 40.E-5 
*----------------------------------------------------------------Options 
OPTI CSTA 0.5 
  LOG 1 
*--------------------------------------------------Transient calculation 
CALC TINI 0. TFIN 40.E-3 
*=========================================================POST-TREATMENT 
SUIT  
Post-treatment 

ECHO 
RESU ALIC TEMP GARD PSCR 
SORT GRAP 
AXTE 1E3 'Temps (ms)' 
*------------------------------------------------------Curve definitions 
COUR  1 'noeud P1' DEPL COMP 2 NOEU LECT p1 TERM 
COUR  2 'noeud R1' DEPL COMP 2 NOEU LECT r1 TERM 
COUR  3 'noeud S1' DEPL COMP 2 NOEU LECT s1 TERM 
*------------------------------------------------------------------Plots 
trac  3 AXES 1 'Deplacement' yzer 
trac  1 2 3 AXES 1 'Deplacement' yzer 
*--------------------------------------------------Results qualification 
QUAL DEPL COMP 2 REFE 2.38669E-02 TOLE 1.E-2 LECT s1 TERM 
*======================================================================= 
FIN 

 

DOMA01 
 
This calculation uses three alternative domain decompositions (in just one run of the 
code): 

• Two conforming domains (lower mesh) with the same grid as in the reference 
solution; 

• Two non-conforming domains (middle mesh) of which the left one has the 
same element size as the reference, while the right one is coarser. The 
interface between these two domains is therefore non-conforming. 

• Same as the previous case (upper mesh) but the right, coarse sub-domain is 
represented by a modal reduction. 

•  
The mesh generation file, including also the Cast3m command to compute the modes 
in the modal sub-domain, is: 
 
*%siz 50 
* 
OPTI ECHO 1 DIME 2 ELEM QUA4; 
opti titr 'DOMA - 01'; 
opti trac psc ftra 'doma01_mesh.ps'; 
* 
P11=0. 0.; 
P12=0. 1.; 
P13=2. 0.; 
P14=2. 1.; 
* 
P21=0. 2.; 
P22=0. 3.; 
P23=2. 2.; 
P24=2. 3.; 
* 
P31=0. 4.; 
P32=0. 5.; 
P33=2. 4.; 
P34=2. 5.; 
* 
VEC1=2. 0.; 
* 
P1=P11; 
R1=P13; 
S1=P13 PLUS VEC1; 
* 
P2=P21; 
R2=P23; 
S2=P23 PLUS VEC1; 
* 
P3=P31; 
R3=P33; 
S3=P33 PLUS VEC1; 
* 
N1=2; 
N2=3; 
N3=4; 
* 
P11_P12=P11 D N2 P12; 
P11_P12P=P11_P12 PLUS VEC1; 
P13_P14=P13 D N2 P14; 
P13_P14P=P13_P14 PLUS VEC1; 
* 
DOM11=P11_P12 REGL N3 P11_P12P; 
DOM12=P13_P14 REGL N3 P13_P14P; 
* 
P21_P22=P21 D N2 P22; 
P21_P22P=P21_P22 PLUS VEC1; 
P23_P24=P23 D N1 P24; 
P23_P24P=P23_P24 PLUS VEC1; 
* 
DOM21=P21_P22 REGL N3 P21_P22P; 
DOM22=P23_P24 REGL N3 P23_P24P; 
* 
P31_P32=P31 D N2 P32; 
P31_P32P=P31_P32 PLUS VEC1; 
P33_P34=P33 D N1 P34; 
P33_P34P=P33_P34 PLUS VEC1; 
* 
DOM31=P31_P32 REGL N3 P31_P32P; 
DOM32=P33_P34 REGL N3 P33_P34P; 
* 
TRAC (DOM11 ET DOM12 ET DOM21 ET DOM22 ET DOM31 ET DOM32); 
* 
ELIM 0.001 DOM11 (P1 ET R1); 
ELIM 0.001 DOM12 (R1 ET S1); 
ELIM 0.001 DOM21 (P2 ET R2); 
ELIM 0.001 DOM22 (R2 ET S2); 
ELIM 0.001 DOM31 (P3 ET R3); 
ELIM 0.001 DOM32 (R3 ET S3); 
* 
TOUT=DOM11 ET DOM12 ET DOM21 ET DOM22 ET DOM31 ET DOM32; 
* 
OPTI SAUV FORMAT 'doma01.msh'; 
SAUV FORMAT TOUT; 
* 
*======================================================================= 
* Reduction Craig-Bampton sur le domaine 2 du cas 3 
* 
NMOD1=10; 
* 
MOD1=MODELE DOM32 MECANIQUE ELASTIQUE ISOTROPE; 
MAT1=MATERIAU MOD1 YOUN 210.E9 NU 0.3 RHO 7800.; 
* 
CL1=BLOQ DEPL P33_P34; 
* 
MAS1=MASSE MOD1 MAT1; 
RIG1=(RIGIDITE MOD1 MAT1) ET CL1; 
* 

* Calcul des modes statiques 
* 
TDEPI1=TABLE; 
TLREE1=TABLE; 
NBN1=NBNO(P33_P34); 
NSTA1=0; 
* 
REPETER BCL1 (NBN1); 
   SI (&BCL1 EGA 1); 
      LREEL1=PROG 1 * 1. (NBN1-1) * 0.; 
      LREEL2=PROG NBN1 * 0.; 
   SINON; 
      SI (&BCL1 EGA NBN1); 
         LREEL1=PROG (NBN1-1) * 0. 1 * 1.; 
         LREEL2=PROG NBN1 * 0.; 
      SINON; 
         LREEL1=PROG (&BCL1-1) * 0. 1. (NBN1-&BCL1) * 0.; 
         LREEL2=PROG NBN1 * 0.; 
      FINSI; 
   FINSI;   
   REPETER BCL2 (2); 
      REPETER BCL3 (2); 
         TLREE1 . &BCL3=LREEL2; 
      FIN BCL3; 
      TLREE1 . &BCL2=LREEL1; 
      CHIMP1=MANU CHPO P33_P34 2 UX TLREE1 . 1 UY TLREE1 . 2; 
      CHIMP11=REDU DOM32 CHIMP1; 
      NSTA1=NSTA1+1; 
      TDEPI1 . NSTA1=CHIMP11; 
   FIN BCL2; 
FIN BCL1; 
* 
TSTAT1=TABLE; 
* 
REPETER BCL4 (2*NBN1); 
   CHFOR1=DEPI CL1 TDEPI1 . &BCL4; 
   CHSTA1=RESOUD RIG1 CHFOR1; 
   TSTAT1 . &BCL4=CHSTA1; 
   LIST &BCL4; 
FIN BCL4; 
* 
* Modes propres encastres 
* 
TMOD1=VIBR 'SIMULTANE' 10. NMOD1 MAS1 RIG1 IMPR TBAS; 
*       
TDYN1=TABLE; 
* 
REPETER BCL5 (NMOD1); 
   CHMOD1=TMOD1 . MODES . &BCL5 . DEFORMEE_MODALE; 
   TDYN1 . &BCL5=CHMOD1; 
FIN BCL5; 
* 
* Normalisation des modes dynamiques 
* 
REPETER BCL6 (NMOD1); 
   CHPO1=TDYN1 . &BCL6; 
   SCAL1=XTMX CHPO1 MAS1; 
   SCAL1=SCAL1 ** 0.5; 
   CHPO1=CHPO1 / SCAL1; 
   CHPO1=ENLEVER CHPO1 LX; 
   TDYN1 . &BCL6=CHPO1; 
FIN BCL6; 
* 
* On retire aux modes statiques leur contribution sur les modes dynamiques 
* 
REPETER BCL7 (NSTA1); 
   CHPO1=TSTAT1 . &BCL7; 
   CHPO2=TSTAT1 . &BCL7; 
   REPETER BCL8 (NMOD1); 
      CHPO3=TDYN1 . &BCL8; 
      SCAL1=YTMX MAS1 CHPO1 CHPO3; 
      CHPO3=SCAL1 * CHPO3; 
      CHPO2=CHPO2 - CHPO3; 
   FIN BCL8; 
   CHPO2=ENLEVER CHPO2 LX; 
   TSTAT1 . &BCL7=CHPO2; 
   LIST &BCL7; 
FIN BCL7; 
* 
* Orthogonalisation de la famille des modes statiques modifies 
* 
TSTAT2=TABLE; 
REPETER BCL9 (NSTA1); 
   CHPO1=TSTAT1 . &BCL9; 
   CHPO2=TSTAT1 . &BCL9; 
   BOOL1=&BCL9 > 1; 
   SI BOOL1; 
      REPETER BCL10 (&BCL9-1); 
         CHPO3=TSTAT2 . &BCL10; 
         SCAL1=YTMX MAS1 CHPO1 CHPO3; 

         CHPO3=SCAL1 * CHPO3; 
         CHPO2=CHPO2 - CHPO3; 
      FIN BCL10; 
   FINSI; 
   CHPO2=ENLEVER CHPO2 LX; 
   SCAL1=XTMX CHPO2 MAS1; 
   SCAL1=SCAL1 ** 0.5; 
   CHPO2=CHPO2 / SCAL1; 
   TSTAT2 . &BCL9=CHPO2; 
   LIST &BCL9; 
FIN BCL9; 
* 
* Base modale finale 
* 
TBAS1=TABLE 'BASE_MODALE'; 
TVEC1=TABLE 'MODES'; 
TVEC1 . MAILLAGE=TMOD1 . MODES . MAILLAGE; 
REPETER BCL11 (NSTA1); 
   TVEC1 . &BCL11=TABLE 'MODE'; 
   TVEC1 . &BCL11 . DEFORMEE_MODALE=TSTAT2 . &BCL11; 
   TVEC1 . &BCL11 . POINT_REPERE=(0. 0.); 
   TVEC1 . &BCL11 . FREQUENCE=0.; 
   TVEC1 . &BCL11 . MASSE_GENERALISEE=1.; 
FIN BCL11; 
REPETER BCL12 (NMOD1); 
   TVEC1 . (NSTA1 + &BCL12)=TABLE 'MODE'; 
   TVEC1 . (NSTA1 + &BCL12) . DEFORMEE_MODALE=TDYN1 . &BCL12; 
   TVEC1 . (NSTA1 + &BCL12) . POINT_REPERE=(0. 0.); 
   TVEC1 . (NSTA1 + &BCL12) . FREQUENCE=0.; 
   TVEC1 . (NSTA1 + &BCL12) . MASSE_GENERALISEE=1.; 
FIN BCL12; 

TBAS1 . MODES=TVEC1; 
* 
* Projection des matrices de masse et de rigidite 
* 
MASP1=PJBA MAS1 TBAS1; 
RIGP1=PJBA RIG1 TBAS1; 
* 
* Frequence max du systeme reduit 
* 
TSTAB1=VIBR 'INTERVALLE' 0. 50000. HAUTE 1 MASP1 RIGP1 'IMPR' 'TBAS'; 
* 
* Sauvegarde des donnees pour PLEXUS 
* 
NTOT1=NSTA1 + NMOD1; 
TSORT1=TABLE; 
REPETER BCL13 (NTOT1); 
   TSORT1 . &BCL13=TBAS1 . MODES . &BCL13 . DEFORMEE_MODALE; 
FIN BCL13; 
* 
OPTI SAUV FORMAT 'doma01.modes'; 
SAUV FORMAT TOUT TSORT1; 
* 
TDON1=TABLE; 
TDON1 . 1=NSTA1; 
TDON1 . 2=NMOD1; 
TDON1 . 3=TSTAB1 . MODES . 1 . FREQUENCE; 
* 
OPTI SAUV FORMAT 'doma01.matr'; 
SAUV FORMAT TDON1 MASP1 RIGP1; 
* 
FIN; 

The input file is: 
 
DOMA - 01 
*----------------------------------------------------------------------- 
ECHO 
!conv win 
CAST tout 
*-----------------------------------------------------------Problem type 
DPLA NONL 
*-----------------------------------------------------------Dimensioning 
DIME 
  PT2L 110 CAR4 64 CL2D 10 ZONE 2 
  TABL 1 4 ECRO 522 
  FORC 10  MTPO 9 
TERM 
*---------------------------------------------------------------Geometry 
GEOM CAR4 tout  
     CL2D p21_p22p p23_p24 p31_p32p p33_p34 
     TERM 
*--------------------------------------------------Geometric Complements 
COMP GROU 4 
     'interf' LECT p21_p22p p23_p24 p31_p32p p33_p34 TERM 
     'prob1'  LECT dom11 dom12 TERM 
     'prob2'  LECT dom21 dom22 p21_p22p p23_p24 TERM 
     'prob3'  LECT dom31 dom32 p31_p32p p33_p34 TERM 
     NGRO 4 
     'tpln'   LECT p1 r1 s1 p2 r2 s2 p3 r3 s3 TERM 
     'char1'  LECT p23_p24p p33_p34p TERM 
     'char2'  LECT p13_p14p TERM 
     'bloqa'  LECT p11_p12 p21_p22 p31_p32 TERM  
     COUL turq LECT prob1 TERM 
          rose LECT prob2 TERM 
          vert LECT prob3 TERM 
*----------------------------------------------------------Material data 
MATE 
   LINE RO 7800. YOUN 210.E9 NU 0.3 
        LECT tout TERM 
   FANT 0. 
        LECT interf TERM 
*--------------------------------------------------------Applied "loads" 
CHAR 1 FACT 2  
   FORC 2 1.E7   LECT char1 TERM 
   FORC 2 0.75E7 LECT char2 TERM 
   TABL 2 0.0 1.0 1.0 1.0 
*----------------------------------------------------Boundary conditions 
LINK COUP BLOQ 12 LECT bloqa TERM 
*----------------------------------------------------------------Outputs 
ECRI DEPL TFRE 4D-3 NOEL  
          POIN LECT tpln TERM 
     FICH ALIC TEMP TFREQ 0.2D-3 
          POIN LECT tpln TERM 
          ELEM LECT 1 TERM 
     FICH ALIC FREQ 8 
*----------------------------------------------------------------Options 
OPTI CSTA 0.5 
  LOG 1 
*---------------------------------------------------Domain decomposition 
STRU 6 
   DOMA LECT dom11          TERM IDEN 91 
   DOMA LECT dom12          TERM IDEN 92 
   DOMA LECT dom21 p21_p22p TERM IDEN 93 

   DOMA LECT dom22 p23_p24  TERM IDEN 94 
   DOMA LECT dom31 p31_p32p TERM IDEN 95 
   MODA LECT dom32 p33_p34  TERM IDEN 96 
   FICH FORM 9 POST TOUS 
$ 
$--- Attention : pour les BENCHs, le fichier des modes est concatené au  
$                fichier maillage. D'ou la valeur 9, qui est la valeur par 
$                defaut du fichier maillage, affectée au fichier  des modes. 
$                Ce fait vient du fait que l'atelier logiciel ne peut gérer  
$                pour l'instant que 2 fichiers input 
$ 
$    POUR LES CALCULS STANDARDS : ces deux fichiers DOIVENT ETRE SEPARES !!! 
$    Il faut donner un numero logique différent de 9 (unité logique par défaut  
$    du fichier maillage) 
$    Si ce numero n'est pas donné, Europlexus l'affectera le numero 50  
$    et l'extension par defaut du fichier est '.mod' 
$  
INTE 3 
   COMP TOLE 0.001 
      DOMA 91 LECT p11_p12p TERM 
      DOMA 92 LECT p13_p14  TERM 
   OPTI  
      DOMA 93 LECT p21_p22p TERM 
      DOMA 94 LECT p23_p24  TERM 
   MORT  
      DOMA 96 LECT p33_p34  TERM 
      DOMA 95 LECT p31_p32p TERM 
*--------------------------------------------------Transient calculation 
CALC TINI 0. TFIN 40.E-3 
*=========================================================POST-TREATMENT 
SUIT  
Post-treatment 
ECHO 
RESU ALIC TEMP GARD PSCR 
SORT GRAP 
AXTE 1E3 'Temps (ms)' 
*------------------------------------------------------Curve definitions 
COUR  1 'noeud P1' DEPL COMP 2 NOEU LECT p1 TERM 
COUR  2 'noeud R1' DEPL COMP 2 NOEU LECT r1 TERM 
COUR  3 'noeud S1' DEPL COMP 2 NOEU LECT s1 TERM 
COUR 11 'noeud P2' DEPL COMP 2 NOEU LECT p2 TERM 
COUR 12 'noeud R2' DEPL COMP 2 NOEU LECT r2 TERM 
COUR 13 'noeud S2' DEPL COMP 2 NOEU LECT s2 TERM 
COUR 21 'noeud P3' DEPL COMP 2 NOEU LECT p3 TERM 
COUR 22 'noeud R3' DEPL COMP 2 NOEU LECT r3 TERM 
COUR 23 'noeud S3' DEPL COMP 2 NOEU LECT s3 TERM 
*------------------------------------------------------------------Plots 
trac  1 11 21 AXES 1 'Deplacement' yzer 
trac  2 12 22 AXES 1 'Deplacement' yzer 
trac  3 13 23 AXES 1 'Deplacement' yzer 
*--------------------------------------------------Results qualification 
QUAL DEPL COMP 2 REFE 2.38669E-02 TOLE 1.E-2 LECT s1 TERM 
     DEPL COMP 2 REFE 2.44172E-02 TOLE 1.E-2 LECT s2 TERM 
     DEPL COMP 2 REFE 2.81579E-02 TOLE 1.E-2 LECT s3 TERM 
*======================================================================= 
FIN 
 

The tip displacement in the reference case is: 

The tip displacement in the three cases with domain decomposition are: 
 

 
 
 
 
The final displacement field in the reference case and in the three cases with domain 
decomposition are: 
 

 


